# $Id$
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: Tobias Kieslich <tobias@justdreams.de>

pkgname=libpano13
pkgver=2.9.19
pkgrel=2
pkgdesc='Basic library to calculate panoramical pictures - core functions of the panotools'
url='http://panotools.sourceforge.net/'
arch=('x86_64')
license=('GPL2')
depends=('libpng' 'libtiff')
makedepends=('jdk7-openjdk')
source=("https://downloads.sourceforge.net/sourceforge/panotools/${pkgname}-${pkgver}.tar.gz")
sha512sums=('f9f147426d44d8babac09cb1a1b5111f88b9c2fc079f33df80133492bcf24cbd1edd771e6bec7bed4b0256bf15c700139c0a83119d3adf296ae5c567e006d868')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	./configure --prefix=/usr --mandir=/usr/share/man --with-java=${JAVA_HOME}
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}
